現場のための水理学 (1)

——単断面における不等流計算——

中津川　誠　清水康行

はじめに

我々河川技術者が，実用面で水理学を必要とする場合，どうも水理学のセオリーはややこしいとか，専門書を見ても，真に必要となるポイントをつかむことに苦労するとかに辟易して，実際は根底にある考え方をあまり省みることなく，また検証することなく，結果のみをうのみにしている例はままあるのではないかと思います。

しかしながら，やはりでてきた結果について，その根拠や経緯を知り，また，自らも手を下して縦横無尽に解を導きだせれば，技術者としての向上や視野の拡大に通ずるところもあるのではないかと思います。

また，最近になって，２次元の数値計算により，河川の流れや河床変動が良好に再現できるようなモデルが開発されてきております。このようなモデルも，水理学の基礎的な知見を営々と積重ねたものに過ぎません。したがって，本稿で解説する基礎知識を着実にマスターしていくことが，一見難解にみえるセオリーを理解していくことにもなるのです。

以上のようなことを踏まえ，もっぱら実用性が高く，かつ技術者の啓蒙に資する意味から“真に必要となる内容を抽出し，かつなるべくやさしく”解説し，流れや河床変動といった水理学の実際を習得していこうというのが本稿の目的であります。特に，実践面での充実を目指すため，多数の演習問題を掲載しておきました。これらについては解説や解答を一読するだけでなく，是非とも自らが手を動かし，計算機を駆使して取組むことが，なによりも理解への早道と思われます。

なお，この現場のための水理学は今回を含め，今後，以下の内容をシリーズで掲載していきたいと思います。

第１回　単断面における不等流計算
第２回　一般断面における不等流計算
第３回　掃流砂と河床変動
第４回　浮遊砂と河床変動
第５回　河床変動計算の応用

1. 流れの概念

私たちが川の流れを考えるとき，どのような光景が思い浮かぶでしょうか。ある場所では狭くなったり，広くなったり，あるいは深くなったり，浅くなったり，流れは実にバラエティーに富んでいることがわかると思います。このように，場所によって速さや深さが変わるような流れのことを 不等流 といい，これが今後，実際の水理現象を説明していく際に根幹となるものです。なお，流れをもっと厳密に追求していくと，時間的に変化するものが考えられ，これを不定流といいます。洪水流や感潮河川の流れが，代表的な不定流の例であります。しかしながら，一般的な河川の流れは，時間的な変化を考えない 定流 とみなしてもさしつかえなく，かつその方が流れの性質をやさしく捉えやすいことから，以後は原則として，定流について考えていくことにします。

さて，一見して複雑な流れをどのように表現していけばよいでしょうか。簡単にいうと，ある地点に水が流れるとき，これを示すのに必要なのは，流体のおかれている位置の高さ (河床高)，水深そして流速の３つに過ぎません。これらをもって，流れを表示したのが図-1.1です。この図でわかるように，流れを構成する３つの要素はすべて長さの次元，つまり，水頭をもって表現していることがわかると思います。例えば，流速ならば２乗して $2g$ (g は重力加速度で 9.8 m/s²) で割ってあります。そして，各水頭をたし合わせたものが，そこで流体のもつ総水頭，いい換えるとエネルギーを表わしていること



図—1.1 流れの表示

になります。この地点・地点の総水頭を結んだものが，図中の点線で示される $\boxed{\text{エネルギー線}}$ というものになるわけです。つまり，エネルギー線は各地点における

$$E = z+h+V^2/2g \quad \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots (1.1)$$

ここで，z；河床高 (m)，h；水深 (m)，V；流速 (m/s)

で表わされる E を結んでいったものにほかありません。また，流れの表示上，基準線と各種線の傾きを定義しておく必要があります。図に示されるように，河床の基準線（一般には水平）に対する傾きを $\boxed{\text{河床勾配}}$ といい，i_b で，水面の基準線に対する傾きを $\boxed{\text{水面勾配}}$ といい，i_w で，そしてエネルギー線が基準線となす傾きを $\boxed{\text{エネルギー勾配}}$ といい，i_e で表わすこととします。

2. 不等流計算

2-1 基礎式

不等流の場合，流れの様子はどうなるでしょう。前章で示した表示法に従って，任意の２つの断面間で流れの状態を描いたものが 図-2.1 です。想定した２つの断面を，各々断面1，断面２とし，各断面の諸量には断面番号の下付き数字をつけることとします。また，図中に新たな h_f なる量がでてきますが，これについて説明しておきましょう。図から一見してわかるように，h_f は２つの断面の総水頭差，すなわち，各々の断面で流体がもつエネルギーの差を表わしているわけですが，これは，流体が断面間を流下するときに失なわれるエネルギー（水頭）にほかありません。この損失は，水が流下するとき，流水と川底がこすれたりして生ずる摩擦に由来するところから，h_f のことを $\boxed{\text{摩擦損失水頭}}$ と呼ぶことにします。

以上のことから，任意の２つの断面間では，次式がなりたつことがわかると思います。

$$z_1+h_1+\frac{V_1^2}{2g} = z_2+h_2+\frac{V_2^2}{2g}+h_f \quad \cdots\cdots\cdots\cdots (2.1)$$



図—2.1 不等流の表示

ここで，z；河床高 (m)，h；水深 (m)，V；流速 (m/s)，h_f；摩擦損失水頭 (m)，g；重力加速度 (m/s²)

すなわち，上式が不等流を表わす式であります。そして，この式は水頭に換算されたエネルギーが任意の個所で等しいという，我々をとり巻く大自然を支配する"エネルギー保存の法則"を具現したものの１つとなっているのです。

また，ここで不等流の場合，河床勾配 i_b，水面勾配 i_w およびエネルギー勾配 i_e は，水深や流速が違う分，各々異なるということに注意しておいて下さい。

これをもって，一応，任意断面間でなりたつ不等流の式を導きだすことができたわけですが，これを，より単純な形で表示するため，さらに一般化してみたいと思います。

まず，図-2.1 では，対象とする２つの断面間の距離は l (m) でしたが，これを微少長さ Δx (m) とします。つまり，微少区間を考えることにより，より細かな現象を捉えることを可能とするわけです。

次に，摩擦損失水頭 h_f ですが，これは 図-2.1 を見てもらえば，エネルギー勾配が $i_e=h_f/l$，すなわち，l を Δx とおくと，$i_e=h_f/\Delta x$ となることがわかり，これから，

$$h_f = i_e\cdot\Delta x \quad \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots (2.2)$$

と表わすことができます。

以上のことを念頭において，式(2.1)を書きなおすと，

$$z_1+h_1+\frac{V_1^2}{2g} = z_2+h_2+\frac{V_2^2}{2g}+i_e\cdot\Delta x \quad \cdots\cdots (2.3)$$

上式において，右辺から左辺を引いて，各項を Δx で割ると，

$$\frac{z_2-z_1}{\Delta x}+\frac{h_2-h_1}{\Delta x}+\frac{\dfrac{V_2^2}{2g}-\dfrac{V_1^2}{2g}}{\Delta x}+i_e=0$$
$$\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots (2.4)$$

ここで，Δx をさらに微少量として，すなわち，限りなく０に近い長さとして dx で表わし，また，z や h，$V^2/2g$ の変化量も dz, dh, $d(V^2/2g)$ で表わすと式(2.4)は，

$$\frac{dz}{dx}+\frac{dh}{dx}+\frac{d}{dx}\left(\frac{V^2}{2g}\right)+i_e=0 \quad \cdots\cdots\cdots\cdots (2.5)$$

となります。また，水位 H は，$H=h+z$ であるので，

$$\frac{dH}{dx}+\frac{d}{dx}\left(\frac{V^2}{2g}\right)+i_e=0 \quad \cdots\cdots\cdots\cdots\cdots\cdots (2.6)$$

となります。このように表わした式(2.5)および式(2.6)が，不等流における流れの微少変化を表わした"微分表

示”というものです。

ところで，式(2.5)および式(2.6)にあるエネルギー勾配は，実際にどのようにして算出すればよいでしょうか。この i_e は，下記の式(2.7)を用いて算出することができます。

$$V=\frac{1}{n}R^{2/3}i_e^{1/2} \quad \cdots\cdots (2.7)$$

ここで，V；流速 (m/s)，n；粗度係数，R；径深 (m)，i_e；エネルギー勾配

これが，不等流の平均流速式です。

なお，このような形をもつ平均流速式をマニング型の式といい，今後，頻繁に使われることになります。また，この中にでてくる n を（マニングの）粗度係数といい，河床の粗さや形状などに左右される抵抗を表わすものといわれており，厳密には，きわめて高度な科学的見地より決められねばなりませんが，実用上は，実測値からの逆算や推定などからあらかじめ決められている既知値として扱っていくこととします。さらに，R は径深というもので，流水断面積もしくは流積 A (m^2) を潤辺（断面において水に潤っている辺の長さ）S (m) で割ったもので，

$$R=A/S \quad \cdots\cdots (2.8)$$

と表わされるものです。この辺の概念は，図-2.2をみて把握して下さい。一般の大部分の河川のように，水深に比して川幅が広い場合は，径深 R は水深 h に等しいとみなしてさしつかえありません。すなわち，

$$R\simeq h \quad \cdots\cdots (2.9)$$

となり，このような断面を 広矩形断面 と称しています。

さて，式(2.7)から i_e を求めると，

$$i_e=\left(\frac{nV}{R^{2/3}}\right)^2 \quad \cdots\cdots (2.10)$$

また，平均流速 V は流量 Q (m^3/s) を流積 A (m^2) で割ったものであるので，式(2.10)は，

$$i_e=\left(\frac{nQ}{AR^{2/3}}\right)^2=\frac{n^2Q^2}{A^2R^{4/3}} \quad \cdots\cdots (2.11)$$

となり，これを式(2.6)に代入して，$V=Q/A$ などとおくと，

$$\boxed{\frac{dH}{dx}+\frac{1}{2g}\frac{d}{dx}\left(\frac{Q}{A}\right)^2+\frac{n^2Q^2}{A^2R^{4/3}}=0} \quad \cdots (2.12)$$

となります。この式(2.12)は，最終的に不等流を表わす基礎式となり，今後，頻繁に使われるものなので，確実に覚えておいて下さい。

ここで，特に広矩形断面の場合においては，$R\simeq h$ とおくことができました。また，流積 A は川幅 B(m) と水深 h(m) の積で表わされることから，式(2.12)は，

$$\frac{dH}{dx}+\frac{1}{2g}\frac{d}{dx}\left(\frac{Q}{Bh}\right)^2+\frac{n^2Q^2}{B^2h^{10/3}}=0 \quad (2.13)$$

となり，これが広矩形断面をもつ水路の不等流計算の基礎式となります。



図—2.2 用語の定義

2-2 流れに関する 2，3 の注意

さて，式(2.12)までたどりついてこれを駆使すれば，実際の不等流計算ができるところに到達することができました。しかしながら，その前にもう少し式の表わす流れの性質を，今後のこともあるので吟味しておきたいと思います。

考えやすいように，広矩形断面の不等流，すなわち式(2.13)について調べてみましょう。

H は水位であるので $H=z+h$ となり，左辺第一項は，

$$\frac{dH}{dx}=\frac{d(z+h)}{dx}=\frac{dz}{dx}+\frac{dh}{dx} \quad \cdots\cdots (2.14)$$

ここで，h；水深 (m)，z；河床高 (m)

また，流量 Q は一定，川幅 B も一定とすると，左辺第二項は，

$$\frac{1}{2g}\frac{d}{dx}\left(\frac{Q}{Bh}\right)^2=\frac{Q^2}{2gB^2}\frac{d}{dx}(h^{-2})=-\frac{Q^2}{gB^2h^3}\frac{dh}{dx} \quad \cdots\cdots (2.15)$$

式(2.14)，式(2.15)を式(2.13)に代入すると，

$$\frac{dz}{dx}+\frac{dh}{dx}-\frac{Q^2}{gB^2h^3}\frac{dh}{dx}+\frac{n^2Q^2}{B^2h^{10/3}}=0 \cdots (2.16)$$

これを，dh/dx について整理して解くと，

$$\frac{dh}{dx}=\frac{-\dfrac{dz}{dx}-\dfrac{n^2Q^2}{B^2h^{10/3}}}{1-\dfrac{Q^2}{gB^2h^3}} \quad \cdots\cdots (2.17)$$

また，上流から下流に向かう方向を x の正の向きとすると，$-dz/dx=i_b$（河床勾配）となることから，

$$\frac{dh}{dx}=\frac{i_b-\dfrac{n^2Q^2}{B^2h^{10/3}}}{1-\dfrac{Q^2}{gB^2h^3}} \quad \cdots\cdots (2.18)$$

となり，最終的に一般的な不等流の式を，水深の場所的

変化 (dh/dx) を表現する式として，見方を変えて書きかえることができたわけです。

この式(2.18)をみると，分子＝0すなわち，$dh/dx=0$ となって，水深が場所的に変化しないような場合が考えられますが，このときの水深を $\boxed{\text{等流水深}}$ といいます。これを以下で求めてみましょう。分子＝0より，

$$i_b=\frac{n^2Q^2}{B^2h^{10/3}} \cdots\cdots (2.19)$$

これを h について解いたものが等流水深 h_0 であり，これが，

$$h_0=\left(\frac{n^2Q^2}{B^2i_b}\right)^{3/10}=\left(\frac{nQ}{B\sqrt{i_b}}\right)^{3/5} \cdots\cdots (2.20)$$

で表わされるものです。つまり，等流とはいたる所で水深の等しい特別な流れといえるわけで，このときの水深が式(2.20)で表わされるものなのです。

一方，式(2.18)については，分母＝0，すなわち $dh/dx=\pm\infty$ となるような場合が考えられます。このような場合の水深を，$\boxed{\text{限界水深}}$ といいます。分母＝0より，

$$1=\frac{Q^2}{gB^2h^3} \cdots\cdots (2.21)$$

となり，これを h について整理し，

$$h_c=\sqrt[3]{\frac{Q^2}{gB^2}} \cdots\cdots (2.22)$$

で示されるものです。

そして，$h>h_c$ の流れを $\boxed{\text{常流}}$，$h=h_c$ の流れを $\boxed{\text{限界流}}$，また，$h<h_c$ の流れを $\boxed{\text{射流}}$ と定義します。

ところで，式(2.22)については，$Q=Bh_cV$ を代入して，両辺を3乗して整理すると，

$$h_c^3=\frac{B^2h_c^2V^2}{gB^2} \text{ より } \quad h_c=\frac{V^2}{g} \cdots\cdots (2.23)$$

ここで，V；平均流速 (m/s)

ゆえに，このとき，

$$\frac{V}{\sqrt{gh_c}}=1 \cdots\cdots (2.24)$$

となることがわかります。左辺の流速を重力加速度×水深の平方根で割った数を，特に $\boxed{\text{フルード数}}$ と称することとします。つまり，この定義に従えば，式(2.24)で示されるように，限界流では F_r (フルード数) は1となるわけです。また，常流は $F_r<1$，射流では $F_r>1$ となるわけですが，これについては，各自証明してみて下さい。なお，これらのことを表-2.1にまとめておきます。

表—2.1 常流・射流の区分

常　　流	$h>h_c$，$F_r<1$
限　界　流	$h=h_c$，$F_r=1$
射　　流	$h<h_c$，$F_r>1$

* h_c；限界水深，F_r；フルード数

ここにいたって，常流，射流という流れの区別ができたわけですが，このような区別が意味するところは，“ある水深（限界水深）を境に性質の異なる流れの形態が存在する”ということになるでしょう。そして，この違いは射流の場合，下流側でなんらかの乱れが生じてもその影響を受けないが常流ではそのような影響が上流に及ぶという流れの性質に由来して生じているのです。このような性質は，後で解説する計算の手法上においても，重要なものなので覚えておいて下さい。

2-3 広矩形単断面の不等流計算

それでは，いよいよ実際の不等流計算に臨んでみることにしましょう。初めて計算を行う人のために，本節では不等流計算の中で最も初歩的な広矩形単断面の例から考えていきたいと思います。

もう一度，図-2.1および式(2.13)を見て下さい。対象とする最小の計算区間を図-2.1のように，上流側断面1，下流側断面2の間と設定します。ここで，式(2.13)を実際の計算に用いることができるように，諸量の差をもって表わすこととします。ここで注意しておきたいのは流量 Q は一定としていること，第三項目は断面1での値と断面2での値の平均となっていることです。

$$\frac{H_2-H_1}{\Delta x}+\frac{Q^2}{2g\cdot\Delta x}\left\{\left(\frac{1}{B_2h_2}\right)^2-\left(\frac{1}{B_1h_1}\right)^2\right\} + \frac{n^2Q^2}{2}\left\{\left(\frac{1}{B_2^2h_2^{10/3}}\right)+\left(\frac{1}{B_1^2h_1^{10/3}}\right)\right\}=0 \cdots\cdots (2.25)$$

ゆえに，両辺に Δx をかけ，左辺に下流側，右辺に上流側の諸量を示す項をもってくると，

$$\left[H_2+\frac{Q^2}{2gB_2^2h_2^2}+\frac{n^2Q^2\Delta x}{2B_2^2h_2^{10/3}}\right]=\left[H_1+\frac{Q^2}{2gB_1^2h_1^2} - \frac{n^2Q^2\Delta x}{2B_1^2h_1^{10/3}}\right] \cdots\cdots (2.26)$$

また，水位 H を河床高 z と水深 h の和で表わすと，

$$\underbrace{\left[z_2+h_2+\frac{Q^2}{2gB_2^2h_2^2}+\frac{n^2Q^2\Delta x}{2B_2^2h_2^{10/3}}\right]}_{\text{下流側}} = \underbrace{\left[z_1+h_1+\frac{Q^2}{2gB_1^2h_1^2}-\frac{n^2Q^2\Delta x}{2B_1^2h_1^{10/3}}\right]}_{\text{上流側}} \quad (2.27)$$

となり，このような表わし方を差分表示といって，今後，基礎方程式を実際に計算機を用いて解く場合，広く使われる手法です。

次に，式(2.27)の中でなにが既知量か，なにが未知量かを考えてみましょう。まず，流量Q，河床高z，河幅B，断面間の距離Δxは既知量として与えられます。また，重力加速度g，粗度係数nも定数として与えられます。結局，水深hが未知量として求めるべきものとなりますが，式が1本しかないのに，上流側の水深h_1と下流側の水深h_2を未知量として，両方同時に求めることはできません。そこで，どちらか一方の水深が既知量として与えられなければならないのですが，ここにいたって，2-2節で示された流れの性質が問題となってくるのです。すなわち常流の場合，流れの変化は下流から上流に及び，射流ではこれが上流方向に及ばないため，常に流れは上流から下流に変化します。この事実を計算にも適用し，常流の場合下流から上流方向へ，射流の場合上流から下流方向へ計算を進めていくことにします。つまり，常流の場合下流側のh_2が既知量として与えられ，上流側のh_1を求めていけばよいわけです。なお，断面がいくつかある場合は，下流端の水深が境界条件として与えられ，上流に向かって断面間の計算を逐次行っていくような方法をとることになります。

それでは準備が整いましたので，今後は実際に演習問題をとおして理解を深めてもらうことにしましょう。

(演習問題1)

粗度係数$n=0.02$，河床勾配$i_b=1/1000$，河幅$B=200$ mの広矩形断面に，流量$Q=2000$ m^3/sが常流で流下するときの水面形を求めよ。ただし，下流端の水深を5 m，河床高を0 mとし，$\Delta x=500$ mのピッチで上流5 km地点まで計算すること。

〔演習問題1の解答〕

(1) 考え方

常流では下流側水深h_2を既知とし，上流側水深h_1を



図—2.3　演習問題1における流れ

求めることに留意して，式(2.27)を参考に下式を定義しておく。

$$f(h_1)=h_1+\frac{AA}{h_1^2}+\frac{BB}{h_1^{10/3}}+CC \cdots\cdots\cdots\cdots\cdots\cdots ①$$

ここで，

$$AA=\frac{Q^2}{2gB^2},\quad BB=-\frac{n^2Q^2\Delta x}{2B^2}$$

$$CC=z_1-\left(z_2+h_2+\frac{Q^2}{2gB^2h_2^2}+\frac{n^2Q^2\Delta x}{2B^2h_2^{10/3}}\right)$$

すなわち，①は$f(h_1)=0$となるとき，式(2.27)と同一になる。つまり，$f(h_1)=0$を満たすようなh_1が断面間の関係を満足する上流側水深の正しい値といえるわけである。

$f(h)=0$となるhを求める方法としては種々の方法があるが，今回はニュートン法を用いた。以下にその説明を述べる。

i) h_1になんらかの仮定値を代入し，$f(h_1)$を計算する。

ii) $f(h_1)>\varepsilon$のとき，$h_1=h_1+\Delta h_1$なる更新を行う。ただし，Δh_1は上記のような更新を行ったときに$f(h_1+\Delta h_1)$を0とするようなものを採用しなければならない。そこで，$f(h_1+\Delta h_1)$を一階微分の項までテイラー展開し，

$$f(h_1+\Delta h_1)\fallingdotseq f(h_1)+f'(h_1)\,\Delta h_1=0 \cdots\cdots\cdots\cdots\cdots\cdots ②$$

これから，

$$\Delta h_1=-\frac{f(h_1)}{f'(h_1)} \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots ③$$

とすればよい。

本問の場合，$f'(h)$は①式を微分することにより，

$$f'(h_1)=1-\frac{2AA}{h_1^3}-\frac{10}{3}\frac{BB}{h_1^{13/3}} \cdots\cdots\cdots\cdots\cdots ④$$

で求められる。

iii) ii)で更新したh_1をもって，再度$f(h_1)$の計算を行う。

iv) $f(h_1)$が十分零に近づくまで，すなわち，$|f(h_1)|<\varepsilon$となるまで上記の手順を繰返す。なお，εは打切り誤差といい，解の精度を勘案して十分零に近い値，例えばmmの精度で解をだすなら$\varepsilon=0.001$というような値をとるものである。

(2) 実際の計算

(1)で示した考え方をもとに，以下に1断面目を例として，計算の過程を示す。

i) 流量$Q=2000$ m^3/s，粗度係数$n=0.02$，重力加速度$g=9.8$ m/s^2，水路幅$B=200$ m，河床勾配$i_b=$

表—2.2　計算の経緯　$\varepsilon=0.0001$

回数	h_1	$f(h_1)$	判定	$f'(h_1)$	Δh_1
1	5.00	0.40643	$\lvert f(h_1)\rvert>\varepsilon$	0.94956	-0.42802
2	4.572	0.00215	$\lvert f(h_1)\rvert>\varepsilon$	0.93919	-0.00229
3	4.570	0.00000	$\lvert f(h_1)\rvert<\varepsilon$		

1/1000，下流側水深 $h_2=5.0$ (下流端水深)，下流側の河床高 $z_2=0$ m，上流側の河床高 $z_1=0.5$ m，断面間距離 $\Delta x=500$ m といった諸条件を設定する。なお，打切り誤差 ε は 0.0001 とする。

ii)　①式の AA, BB を求める。

$$AA=\frac{Q^2}{2gB^2}=\frac{2000^2}{2\times9.8\times200^2}=5.10204$$

$$BB=\frac{-n^2\cdot Q^2\cdot\Delta x}{2B^2}=\frac{-0.02^2\times2000^2\times500}{2\times200^2}=-10.0$$

iii)　①式の CC を求める。

$$CC=z_1-\left(z_2+h_2+\frac{AA}{h_2^2}-\frac{BB}{h_2^{10/3}}\right)$$
$$=0.5-\left(0.0+5.0+\frac{5.10204}{5^2}-\frac{-10.0}{5^{10/3}}\right)$$
$$=-4.75087$$

iv)　上流側水深 h_1 を $h_1=h_2=5.0$ m と仮定する。

v)　①式の $f(h_1)$ を求める。

$$f(h_1)=h_1+\frac{AA}{h_1^2}+\frac{BB}{h_1^{10/3}}+CC$$
$$=5.0+\frac{5.10204}{5^2}+\frac{-10.0}{5^{10/3}}-4.75087$$
$$=0.40643$$

vi)　$|f(h_1)|>\varepsilon\,(=0.0001)$ なので，h_1 の更新を行う。まず，④式より $f'(h)$ を求める。

$$f'(h_1)=1-2\cdot\frac{AA}{h_1^3}-\frac{10}{3}\,\frac{BB}{h_1^{13/3}}$$
$$=1-2\cdot\frac{5.10204}{5^3}-\frac{10}{3}\,\frac{(-10.0)}{5^{13/3}}$$
$$=0.94956$$

vii)　③式の Δh_1 を求める。

$$\Delta h_1=-\frac{f(h_1)}{f'(h_1)}$$
$$=-\frac{0.40643}{0.94956}$$
$$=-0.42802$$

viii)　$h_1=h_1+\Delta h_1$ で h_1 を更新し，$f(h_1)$ を求める。

$$h_1=5.0-0.42802=4.57198$$
$$f(h_1)=0.00215$$

ix)　$|f(h_1)|<\varepsilon\,(=0.0001)$ ならばこのときの h_1 が正解となるが，$|f(h_1)|>\varepsilon$ なので vi) にいく。vi)～viii) を計算して，

$$f'(h_1)=0.93919$$
$$\Delta h_1=-0.00229$$
$$h_1=4.56969$$

$f(h_1)=0.00000$ となる。

$|f(h_1)|<\varepsilon$ なので，このときの h_1 である 4.570 が正解。

なお，収束にいたるまでの h_1, $f(h_1)$, $f'(h_1)$, Δh_1 の計算経緯を 表-2.2 に示す。

(3)　計算機プログラムの概要

後述の補遺〔1〕参照。

表—2.3　計算結果

I	$H1$	$H2$	F	N
1	5.00000	5.00000	0.00000	0
2	4.56969	5.06969	0.00000	2
3	4.16940	5.16940	0.00000	2
4	3.81317	5.31317	−0.00000	2
5	3.51830	5.51830	0.00000	2
6	3.29982	5.79982	0.00000	2
7	3.16044	6.16044	0.00005	1
8	3.08495	6.58495	−0.00002	1
9	3.04955	7.04955	−0.00001	1
10	3.03444	7.53444	−0.00000	1
11	3.02829	8.02829	−0.00000	1
12	3.02584	8.52584	0.00000	1
13	3.02488	9.02488	0.00000	1
14	3.02450	9.52450	0.00000	1
15	3.02435	10.02435	0.00000	1
16	3.02429	10.52429	0.00000	1
17	3.02427	11.02427	0.00000	1
18	3.02426	11.52426	−0.00000	1
19	3.02426	12.02426	0.00000	1
20	3.02425	12.52425	0.00000	1

(4) 計 算 結 果

出力結果 (表-2.3) の I は断面 No, $H1$ が水深, $H2$ が水位, F が打切り誤差, N が繰返し計算回数を示している。

今回の計算では, No. 13 断面から水深がほぼ一定となっているのがわかる。これは等流水深に漸近しているためである。

ちなみに, 等流水深 h_0 は,

$$h_0=\left(\frac{Qn}{B\sqrt{i_b}}\right)^{3/5}$$

$$=\left(\frac{2000\times0.02}{200\times\sqrt{\dfrac{1}{1000}}}\right)^{3/5}$$

$=3.024$ m である。

以上, 解答作成者　渡辺和好

〔演習問題 2〕

粗度係数 $n=0.02$, 河床勾配 $i_b=1/100$, 河幅 $B=200$ m の広矩形断面に, 流量 $Q=2000$ m³/s が射流で流下するときの水面形を求めよ。ただし, 上流端の水深を 1.4 m, 河床高を 50 m とし, $\Delta x=500$ m と $\Delta x=100$ m の 2 とおりのピッチで下流 5 km 地点まで計算すること。

〔演習問題 2 の解答〕

(1) 考 え 方

射流では上流から下流に向かって計算を進めるので, 上流側水深 h_1 を既知とし, 下流側水深 h_2 を求めなければならない。

すなわち, 式(2.27)から得られる次式を満たすような h_2 を求めることが必要となる。

$$f(h_2)=h_2+\frac{AA}{h_2^2}+\frac{BB}{h_2^{10/3}}+CC=0 \quad \cdots\cdots\cdots\cdots①$$

ただし,



図—2.4　演習問題 2 における流れ

$$AA=\frac{Q^2}{2gB^2},\quad BB=\frac{\Delta x}{2}\cdot\frac{n^2Q^2}{B^2}$$

$$CC=z_2-\left(z_1+h_1+\frac{AA}{h_1^2}-\frac{BB}{h_1^{10/3}}\right)$$

本問では, ニュートン法による計算法を試みることとする。手順は前問 (演習問題 1) に示したものとおおむね同じであるのでそれを参照されたいが, 若干の追加があるので説明する。

すなわち, h を更新する際求める Δh については, 前問においては更新後の $f(h+\Delta h)$ をテイラー展開したものから,

$$f(h+\Delta h)\doteqdot f(h)+f'(h)\,\Delta h+\frac{1}{2}f''(h)\,\Delta h^2+\frac{1}{6}f'''(h)\,\Delta h^3+\cdots=0 \quad \cdots\cdots\cdots\cdots②$$

一階微分項までをとり,

$$f(h+\Delta h)\doteqdot f(h)+f'(h)\,\Delta h=0 \quad \cdots\cdots\cdots\cdots③$$

$$\Delta h=-f(h)/f'(h) \quad \cdots\cdots\cdots\cdots④$$

としたが, ②式の二階微分項までとるものも考えられる。すなわち,

$$f(h+\Delta h)\doteqdot f(h)+f'(h)\,\Delta h+\frac{1}{2}f''(h)\,\Delta h^2=0\cdots⑤$$

として, これを満たす Δh を求めるものである。今後, 便宜的に前者を1階のニュートン法, 後者を2階のニュートン法と呼称することとする。

前者については, 前問で解説したので説明は省略する。以後は, 後者について説明していく。

2階のニュートン法における更新値 Δh を求めることは容易である。すなわち, ⑤式を便宜的に次式のように表わすと,

$$g(\Delta h)=a\cdot\Delta h^2+b\Delta h+c=0 \quad \cdots\cdots\cdots\cdots⑥$$

ただし, $a=\dfrac{1}{2}f''(h)$, $b=f'(h)$, $c=f(h)$

ゆえに, 2次方程式の解より,

$$\Delta h=\frac{-b\pm\sqrt{b^2-4ac}}{2a} \quad \cdots\cdots\cdots\cdots⑦$$

となる。

しかしながら, ここで1つ問題が生ずる。それは, ⑥式から得られる解は ⑦式に示すように2個あるが, そのいずれを採用するかということである。したがって, 各々の解の性質を調べてみることにする。今, 次のように Δh_+, Δh_- を定義し,

$$\Delta h_+=\frac{-b+\sqrt{b^2-4ac}}{2a} \quad \cdots\cdots\cdots\cdots⑧$$

$$\Delta h_- = \frac{-b-\sqrt{b^2-4ac}}{2a} \cdots\cdots ⑨$$

とすると，分子は $(-b+\sqrt{b^2-4ac})>(-b-\sqrt{b^2-4ac})$ となるので，$a>0$ のとき $\Delta h_+>\Delta h_-$，$a<0$ のとき $\Delta h_+<\Delta h_-$ となり，これを図に表わすと図-2.5のようになる。

これからわかるように，Δh_+ は常に $g(\Delta h)$ の傾きが正のとき，Δh_- は常に $g(\Delta h)$ の傾きが負のときの解である[注1)] ことに注目されたい。

一方，$f(h)$ については，①式にでてくる Δx が非常に小さいとすると，$BB\simeq 0$ となるから，

$$f(h)=h+\frac{AA}{h^2}+CC \cdots\cdots ⑩$$

これをグラフに表わしたものが図-2.6である。ここで示される極小値は⑩式を微分して，

$$f'(h)=1-\frac{2AA}{h^3}=0 \cdots\cdots ⑪$$

$$h=\sqrt[3]{2AA} \cdots\cdots ⑫$$

⑫式の AA は①式より，

$$h=\sqrt[3]{\frac{Q^2}{gB^2}} \cdots\cdots ⑬$$

これは限界水深 h_c にほかならない（本文，式(2.22)参照)。

すなわち，h_c を境に図-2.6を常流，射流の領域に区



図—2.5　$\Delta h\sim g(\Delta h)$ の関係



図—2.6

分することができる。ここで明らかなように，$f(h)=0$ を満たす解 h で $f(h)$ の傾きが正のものが常流時の水深，$f(h)$ の傾きが負のものが射流時の水深[注2)] を表わしていることがわかるであろう。

ところで，前掲の $g(\Delta h)$ については，

$$f(h+\Delta h)=(h+\Delta h)+\frac{AA}{(h+\Delta h)^2}+CC$$

$$\fallingdotseq f(h)+f'(h)\Delta h+\frac{1}{2}f''(h)\Delta h^2$$

$$=a\Delta h^2+b\Delta h+c=g(\Delta h) \cdots\cdots ⑭$$

にほかならない。すなわち，$f(h+\Delta h)$ つまり $f(h)$ のもつ性質は，当然 $g(\Delta h)$ のそれと等価なものである。よって，

$g(\Delta h)$ の傾き正 $\longleftrightarrow$ $f(h)$ の傾き正（常流時）

$g(\Delta h)$ の傾き負 $\longleftrightarrow$ $f(h)$ の傾き負（射流時）

ゆえに，注1)，注2) より，

⑧式の $\Delta h_+ \longrightarrow$ 常流の時の更新量
⑨式の $\Delta h_- \longrightarrow$ 射流の時の更新量

これが，2階のニュートン法における更新時の規定である。ややこしい理由は忘れても，上記枠内の事実だけは覚えておく必要がある。

本問の場合は射流なので，2階のニュートン法を使う場合，⑨式で示されるところの Δh_- をもって更新を行うことに注意されたい。

本問を解く際の手順を以下に示す。

i) ①式の h_2 になんらかの仮定値を代入し，$f(h_2)$ を計算する。

ii) $f(h_2)>\varepsilon$（打切り誤差）のとき，$h_2+\Delta h_2$ なる更新を行う。ここで Δh_2 については，

・1階のニュートン法の場合

$f(h_2+\Delta h_2)\fallingdotseq f(h_2)+f'(h_2)\Delta h_2=0$ より

$$\Delta h_2=-f(h_2)/f'(h_2) \cdots\cdots ⑮$$

・2階のニュートン法の場合

$f(h_2+\Delta h_2)\fallingdotseq f(h_2)+f'(h_2)\Delta h_2+\frac{1}{2}f''(h_2)\Delta h_2^2=0$ より

射流であることを考慮し，

$$\Delta h_2=\frac{-f'(h_2)-\sqrt{(f'(h_2))^2-2f(h_2)f''(h_2)}}{f''(h_2)} \cdots ⑯$$

ただし，$f(h_2)$, $f'(h_2)$, $f''(h_2)$ は，

$$f(h_2)=h_2+\frac{AA}{h_2^2}+\frac{BB}{h_2^{10/3}}+CC \cdots\cdots ⑰（①式）$$

$$f'(h_2)=1-\frac{2AA}{h_2^3}-\frac{10}{3}\frac{BB}{h_2^{13/3}} \cdots\cdots ⑱$$

$$f''(h_2)=\frac{6AA}{h_2^4}+\frac{130}{9}\frac{BB}{h_2^{16/3}}\quad\cdots\cdots\cdots\cdots\cdots\cdots\cdots⑲$$

iii) ii) で更新した h_2 をもって，再度 $f(h_2)$ の計算を行う。

iv) $f(h_2)$ が十分零に近づくまで，すなわち，$|f(h_2)|<\varepsilon$ となるまで ii), iii) の手順を繰返す。

(2) 実 際 の 計 算

(1) で示した考え方をもとに，以下に１断面目を例として計算の過程を示す。解説は１階のニュートン法と２階のニュートン法に区別して行う。

・1階のニュートン法

i) 流量 $Q=2000\ \mathrm{m^3/s}$，粗度係数 $n=0.02$，重力加速度 $g=9.8\ \mathrm{m/s^2}$，水路幅 $B=200\ \mathrm{m}$，河床勾配 $i_b=1/100$，上流側水深 $h_1=1.4\ \mathrm{m}$ (上流端水深)，断面間距離 $\Delta x=100\ \mathrm{m}$，上流側河床高 $z_1=50\ \mathrm{m}$，下流側河床高 $z_2=49\ \mathrm{m}$ といった諸条件設定。なお，打切り誤差 ε は 0.001 とする。

ii) 下流側水深 h_2 を $h_2=h_1=1.4\ \mathrm{m}$ と仮定する。

iii) ① 式の $f(h_2)$ を計算する。$f(h_2)=0.303$ (1 回目の計算)

iv) $|f(h_2)|>\varepsilon$ なので，h_2 を更新する必要がある。そこで，⑮ 式より $\Delta h=-f(h_2)/f'(h_2)=0.071$ となり，更新値は $h_2=h_2+\Delta h=1.4+0.071=1.471$ となる。

v) $h_2=1.471$ に対する $f(h_2)$ を計算する。0.030 (2 回目の計算)

vi) $|f(h_2)|>\varepsilon$ なので再度 h_2 を更新する。そこで，⑮ 式より Δh を更新値をもって求めると 0.009 となり，更新値は $h_2=h_2+\Delta h=1.471+0.009=1.480$ となる。

vii) $h_2=1.480$ に対する $f(h_2)$ を計算する。0.00038 (3 回目の計算)

viii) $|f(h_2)|<\varepsilon$ なので，下流側水深の正解は 1.480 m。

・2階のニュートン法

i), ii) は上記と同様なので省略。

iii) ① 式の $f(h_2)$ は 0.303 (1 回目の計算)

iv) $|f(h_2)|>\varepsilon$ なので h_2 を更新。このとき，射流なので平方根の符号が負となる方の解を採用。(⑯ 式) Δh は 0.083 となる。ゆえに，更新値は $h_2=h_2+\Delta h=1.4+0.083=1.483$ となる。

v) $h_2=1.483$ に対する $f(h_2)$ を計算すると -0.010 (2 回目の計算)

vi) $|f(h_2)|>\varepsilon$ なので h_2 を更新，iv) と同様の手続きをとって Δh を計算すると -0.003 となる。ゆえに更新値は，$h_2=h_2+\Delta h=1.483-0.003=1.480$

vii) $h_2=1.480$ に対する $f(h_2)$ を計算すると，$f(h_2)=-0.0001$ (3 回目の計算)

viii) $|f(h_2)|<\varepsilon$ なので，下流側水深の正解は 1.480 m。

以上の計算についての収束状況については，表-2.4 に示しておく。

表中の K は計算回数，$H2$ は下流側水深，$F(H2)$ は $f(h_2)$ (⑰ 式)，FD は $f'(h_2)$ (⑱ 式)，FDD は $f''(h_2)$ (⑲ 式)，DH は更新量 Δh_2 (⑮ 式もしくは ⑯ 式) を表わすものである。

この２つの計算結果で注目していただきたいのは，1回目の更新で得られた h_2 が，1階のニュートン法では 1.471，2階のニュートン法では 1.483 となり，後者の方がより真値に近づいていることである。すなわち，② 式のような近似式を用いる際，多くの項を考慮した方が真値に収束させるのに有利な方法といえる。しかしながら，当然，後者の方が計算が複雑となり，計算時間が長くなることも考えられることに注意すべきである。なお，一般には１階のニュートン法が多用されている。

(3) 計算機プログラムの概要

後述の補遺〔2〕参照。

表—2.4　1 断面目の計算例　($\Delta x=100$ m)

(1) 1階のニュートン法

K=1	H2=1.400	F(H2)=0.30307
	FD=−4.26996	DH=0.07098
K=2	H2=1.471	F(H2)=0.02988
	FD=−3.45798	DH=0.00864
K=3	H2=1.480	F(H2)=0.00038

(2) 2階のニュートン法

K=1	H2=1.400	F(H2)=0.30307	
	FD=−4.26996	FDD=14.69080	DH=0.08276
K=2	H2=1.483	F(H2)=−.01016	
	FD=−3.33962	FDD=11.57420	DH=−.00303
K=3	H2=1.480	F(H2)=−.00001	

表—2.5 計算結果 (Δx=100 m)

(1) 1階のニュートン法

No.	距離 (m)	河床高 (m)	水深 (m)	水位 (m)	繰返し回数
1	0	50.0	1.400	51.400	0
2	100	49.0	1.480	50.480	3
3	200	48.0	1.505	49.505	3
4	300	47.0	1.513	48.513	2
5	400	46.0	1.515	47.515	2
6	500	45.0	1.515	46.515	2
7	600	44.0	1.515	45.515	1
8	700	43.0	1.515	44.515	1
9	800	42.0	1.515	43.515	1
10	900	41.0	1.515	42.515	1
11	1000	40.0	1.515	41.515	1
12	1100	39.0	1.515	40.515	1
		≀			
49	4800	2.0	1.515	3.515	1
50	4900	1.0	1.515	2.515	1
51	5000	0.0	1.515	1.515	1

(2) 2階のニュートン法

No.	距離 (m)	河床高 (m)	水深 (m)	水位 (m)	繰返し回数
1	0	50.0	1.400	51.400	0
2	100	49.0	1.480	50.480	3
3	200	48.0	1.505	49.505	2
4	300	47.0	1.513	48.513	2
5	400	46.0	1.515	47.515	2
6	500	45.0	1.516	46.516	2
7	600	44.0	1.516	45.516	1
8	700	43.0	1.516	44.516	1
9	800	42.0	1.516	43.516	1
10	900	41.0	1.516	42.516	1
11	1000	40.0	1.516	41.516	1
12	1100	39.0	1.516	40.516	1
		≀			
49	4800	2.0	1.516	3.516	1
50	4900	1.0	1.516	2.516	1
51	5000	0.0	1.516	1.516	1

(4) 計 算 結 果

Δx=100 m と 500 m の場合の計算結果を 表-2.5 および 表-2.6 に示す。

Δx が 100 m の場合と 500 m の場合の計算結果を比較すると，100 m の場合は水深が少しずつ大きくなり等流水深に近づく傾向を示す一方，Δx=500 m の場合は波を打つような形で等流水深に近づいていく。

$$\text{等流水深}\ h_0=\left(\frac{n^2Q^2}{B^2 i_b}\right)^{3/10}=1.516\ \text{m}$$

これは，Δx を大きくとりすぎたことに起因する現象である。このように，Δx を大きくとりすぎると不都合が生じることもあるので，不等流計算を行う場合には，Δx のとり方に注意しなければならない。

以上，解答作成者　福田義昭

表—2.6 計算結果 (Δx=500 m)

(1) 1階のニュートン法

No.	距離 (m)	河床高 (m)	水深 (m)	水位 (m)	繰返し回数
1	0	50.0	1.400	51.400	0
2	500	45.0	1.589	46.589	4
3	1000	40.0	1.485	41.485	4
4	1500	35.0	1.531	36.531	3
5	2000	30.0	1.508	31.508	3
6	2500	25.0	1.519	26.519	3
7	3000	20.0	1.514	21.514	2
8	3500	15.0	1.517	16.517	2
9	4000	10.0	1.515	11.515	2
10	4500	5.0	1.516	6.516	2
11	5000	0.0	1.516	1.516	2

(2) 2階のニュートン法

No.	距離 (m)	河床高 (m)	水深 (m)	水位 (m)	繰返し回数
1	0	50.0	1.400	51.400	0
2	500	45.0	1.589	46.589	3
3	1000	40.0	1.485	41.485	3
4	1500	35.0	1.531	36.531	3
5	2000	30.0	1.508	31.508	2
6	2500	25.0	1.519	26.519	2
7	3000	20.0	1.514	21.514	2
8	3500	15.0	1.517	16.517	2
9	4000	10.0	1.515	11.515	2
10	4500	5.0	1.516	6.516	2
11	5000	0.0	1.516	1.516	2

(演習問題3)

河床高および河幅が表–2.7に示されるような広矩形断面河川に，流量 $Q=1{,}500\ \mathrm{m^3/s}$ が流下した場合の水面形を求めよ。ただし，粗度係数 $n=0.025$，下流端の水深を 2.5 m とする。

表—2.7 演習問題3における河道諸元

断面番号	下流端からの距離 (m)	河床高 (標高) z (m)	河幅 B (m)
1	0	0	300
2	500	0.5	320
3	1000	0.9	280
4	1200	0.8	250
5	1800	2.0	300
6	2100	2.3	300
7	2500	3.0	320
8	3000	3.0	350
9	3300	3.5	300
10	3800	4.0	250

〔演習問題3の解答〕

(1) 考え方

考え方は，演習問題1とほとんど同じである。ただし，断面によって河幅が違うこと。また，河床勾配も一様でないことに注意する。常流であるから，下流から上流に向かって計算するので，下流側水深 h_2 を既知とし，上流側水深 h_1 を求めることから，式(2.27)を参考として，

$$f(h_1)=h_1+\frac{AA}{h_1^2}+\frac{BB}{h_1^{10/3}}+CC \cdots\cdots ①$$

ただし，

$$AA=\frac{Q^2}{2gB_1^2} \cdots\cdots ②$$

$$BB=-\frac{n^2Q^2\Delta x}{2B_1^2} \cdots\cdots ③$$

$$CC=z_1-\left(z_2+h_2+\frac{Q^2}{2gB_2^2h_2^2}+\frac{n^2Q^2\Delta x}{2B_2^2h_2^{10/3}}\right) \cdots\cdots ④$$

本問では，ニュートン法 (1階および2階) を用いることとする。手順は前問 (演習問題1および2) を参照されたい。

注意事項としては，

- 河幅が断面ごとに異なる。
- 2階のニュートン法の場合，更新量 Δh に注意 (演習問題2参照) する。

の2点である。

(2) 実際の計算

・1階のニュートン法

i) 流量 $Q=1500\ \mathrm{m^3/s}$，粗度係数 $n=0.025$，重力加速度 $g=9.8\ \mathrm{m/s^2}$，上流側水路幅 $B_1=320$ m，下流側水路幅 $B_2=300$ m，下流側水深 $h_2=2.5$ m，上流側河床高 $z_1=0.5$ m，下流側河床高 $z_2=0$ m，断面間距離 $\Delta x=500$ m といった諸条件設定。なお，打切り誤差 ε は 0.001 とする。

ii) 上流側水深

h_1 の初期値として 2.5 m を代入する。

iii) ①式の係数 AA, BB, CC を計算する。

$AA=Q^2/(2gB_1^2)=1{,}500^2/(2\times9.8\times320^2)=1.12105$

$BB=-Q^2n^2\Delta x/(2B_1^2)=-1{,}500^2\times0.025^2\times500/(2\times320^2)=-3.43323$

$$CC=z_1-\left\{z_2+h_2+\frac{Q^2}{2gB_2^2h_2^2}+\frac{\Delta xn^2Q^2}{2B_2^2h_2^{10/3}}\right\}$$
$$=0.5-\left\{0+2.5+\frac{1500^2}{2\times9.8\times300^2\times2.5^2}+\frac{500\times0.025^2\times1500^2}{2\times300^2\times2.5^{10/3}}\right\}=-2.38828$$

iv) $f(h_1)$ を計算し $|f(h_1)|>\varepsilon$ ならば，h_1 を以下の手順をもって更新する。

$$\begin{cases} f(h_1)=h_1+\dfrac{AA}{h_1^2}+\dfrac{BB}{h_1^{10/3}}+CC \quad (①式より) \\ f'(h_1)=1-\dfrac{2AA}{h_1^3}-\dfrac{10}{3}\dfrac{BB}{h_1^{13/3}} \\ \Delta h_1=-\dfrac{f(h_1)}{f'(h_1)} \\ h_1=h_1+\Delta h_1 \end{cases}$$

v) $|f(h_1)|<\varepsilon$ となるまで iv) の手順を繰返す。

$|f(h_1)|<\varepsilon$ となった時点で，上下流のエネルギー差

表—2.8 $\varepsilon=0.001$

回数	水深 (m)	$f(h_1)$	判定	$f'(h_1)$	Δh_1
0	2.5	0.12919	$\lvert f(h_1)\rvert>\varepsilon$	1.07237	−0.1205
1	2.3795	−0.00166	$\lvert f(h_1)\rvert>\varepsilon$	1.10097	0.0015
2	2.381	-9.8×10^{-6}	$\lvert f(h_1)\rvert<\varepsilon$		

が 0.001 未満になったということで，計算を打切る。

なお，iv), v) の経緯については表-2.8に示しておく。

・2階のニュートン法

i), ii), iii) は1階のニュートン法と同じ。

iv) $f(h_1)$ を計算し $|f(h_1)|>\varepsilon$ ならば，h_1 を以下の手順をもって更新。ただし，$\varDelta h_1$ については常流であるので，平方根の前の符号を正とする（演習問題2参照）。

$$\begin{cases} f(h_1)=h_1+\dfrac{AA}{h_1^2}+\dfrac{BB}{h_1^{10/3}}+CC \quad (\text{①式より}) \\ f'(h_1)=1-\dfrac{2AA}{h_1^3}-\dfrac{10}{3}\dfrac{BB}{h_1^{13/3}} \\ f''(h_1)=\dfrac{6AA}{h_1^4}+\dfrac{130}{9}\dfrac{BB}{h_1^{16/3}} \\ \varDelta h_1=\dfrac{-f'(h_1)+\sqrt{\{f'(h_1)\}^2-2\cdot f''(h_1)\cdot f(h_1)}}{f''(h_1)} \\ h_1=h_1+\varDelta h_1 \end{cases}$$

v) $|f(h_1)|<\varepsilon$ となるまで iv) の手順を繰返す。

なお，iv), v) の経緯については表-2.9に示しておく。

以上の結果から，1階のニュートン法と2階のニュートン法を比較すると，収束回数については大差ないことがわかった。ゆえに，今回のような計算を行う場合，1階のニュートン法を用いて計算を行った方が計算も複雑とならず，実用的といえる。

(3) 計算機プログラムの概要

後述の補遺〔3〕参照。

(4) 計 算 結 果

表-2.10に，1階のニュートン法および2階のニュートン法の結果を掲載する。

以上，解答作成者　村上泰啓

表—2.9　　$\varepsilon=0.001$

回　　数	水　深 (m)	$f(h_1)$	判　　定	$f'(h_1)$	$f''(h_1)$	$\varDelta h$
0	2.5	0.12919	$\lvert f(h_1)\rvert>\varepsilon$	1.07237	−0.20197	−0.11914
1	2.3809	−0.00012	$\lvert f(h_1)\rvert<\varepsilon$			

表—2.10　計 算 結 果

(1) 1階のニュートン法

No.	追加距離	区間距離	河　幅	河床高	水　深	水　位	収束回数
1	0.00	0.00	300.0	0.0000	2.5000	2.5000	0
2	500.00	500.00	320.0	0.5000	2.3810	2.8810	2
3	1000.00	500.00	280.0	0.9000	2.3622	3.2622	1
4	1200.00	200.00	250.0	0.8000	2.6529	3.4529	2
5	1800.00	600.00	300.0	2.0000	2.0860	4.0860	3
6	2100.00	300.00	300.0	2.3000	2.1872	4.4872	2
7	2500.00	400.00	320.0	3.0000	1.9798	4.9798	2
8	3000.00	500.00	350.0	3.0000	2.5980	5.5980	2
9	3300.00	300.00	300.0	3.5000	2.2138	5.7138	2
10	3800.00	500.00	250.0	4.0000	2.2610	6.2610	1

計算時間 (SEC)＝00 : 00 : 02
収束回数計　　＝17

(2) 2階のニュートン法

No.	追加距離	区間距離	河　幅	河床高	水　深	水　位	収束回数
1	0.00	0.00	300.0	0.0000	2.5000	2.5000	0
2	500.00	500.00	320.0	0.5000	2.3809	2.8809	1
3	1000.00	500.00	280.0	0.9000	2.3622	3.2622	1
4	1200.00	200.00	250.0	0.8000	2.6533	3.4533	1
5	1800.00	600.00	300.0	2.0000	2.0862	4.0862	2
6	2100.00	300.00	300.0	2.3000	2.1874	4.4874	1
7	2500.00	400.00	320.0	3.0000	1.9798	4.9798	2
8	3000.00	500.00	350.0	3.0000	2.5983	5.5983	2
9	3300.00	300.00	300.0	3.5000	2.2140	5.7140	2
10	3800.00	500.00	250.0	4.0000	2.2617	6.2617	1

計 算 時 間　(SEC)＝00：00：03
収束回数計　　　＝13

補遺　計算プログラム概要

〔1〕 演習問題 1

(1) プログラムのフローチャート



(2) プログラムの解説

文 番 号	解　　説
7～17	初期値の設定，$H1$，$H2$，HH：水深，HHH：水位，G：重力加速度，Q：流量，B：河幅，AN：粗度，DX：断面間距離，Z：河床高
22	①式の AA の計算．
23	①式の BB の計算．
26	①式の CC の計算．
29	$f(h)$ の計算．
32	$f'(h)$ の計算．
34	Δh の計算．
36	h の更新
37	$f(h)$ の計算．
41	収束の判断，収束したら 99 へいく．収束していないならば 88（32 行目）にいく．
43～48	次の断面における条件設定．
55～58	計算結果の出力．

(3) プログラムのリスト

```
      C     FUTORYU KEISAN PROGRAM
      C*****  NEWTON-HOU  *****
            DIMENSION HH(30),JJ(30),FF(30),HHH(30)
      C
      C     SHOKI JYOUKEN
      C
            H1=5.0
            H2=5.0
            HH(1)=5.0
            HHH(1)=5.0
            G=9.8
            Q=2000.0
            B=200.0
            AN=0.02
            DX=500.0
            Z1=0.5
            Z2=0.0
            J=0
      C
      C     KEISAN KAISHI
      C
            AA=Q**2/(2*G*B**2)
            BB=-AN**2*Q**2*DX/(2*B**2)
            WRITE(6,21)AA,BB
         21 FORMAT(1H ,"AA=",F10.5,5X,"BB=",F10.5)
         66 CC=Z1-(Z2+H2+AA/H2**2-BB/H2**(10./3.))
            J=J+1
            J1=0
            FH=H1+AA/H1**2+BB/H1**(10./3.)+CC
            WRITE(6,22)J,CC,FH
         22 FORMAT(1H ,"J=",I5,5X,"CC=",F10.5,5X,"FH=",F10.5)
         88 DFH=1-2*AA/H1**3-10./3.*BB/H1**(13./3.)
            J1=J1+1
            DH=-FH/DFH
            H0=H1
            H1=H1+DH
            FH=H1+AA/H1**2+BB/H1**(10./3.)+CC
            WRITE(6,23)J1,FH,DFH,DH,H0,H1
         23 FORMAT(1H ,"J1=",I5,5X,"FH=",F10.5,5X,"DFH=",F10.5,5X,
           *"DH=",F10.5,5X,"H0=",F10.5,5X,"H1=",F10.5)
            IF(ABS(FH).LT.0.0001)GO TO 99
            GO TO 88
         99 HH(J+1)=H1
            HHH(J+1)=HH(J+1)+Z1
            Z1=Z1+0.5
            Z2=Z2+0.5
            JJ(J+1)=J1
            FF(J+1)=FH
            IF(J+1.GT.20)GO TO 55
            H2=H1
            GO TO 66
      C
      C     KEISAN KEKKA NO SHUTURYOKU
      C
         55 WRITE(6,11)
            WRITE(6,12)(I,HH(I),HHH(I),FF(I),JJ(I),I=1,20)
         11 FORMAT(1H1,5X,"I",10X,"H1",11X,"H2",10X,"F",10X,"N")
         12 FORMAT(1H ,4X,I2,7X,F10.5,3X,F10.5,3X,F10.5,2X,I2)
            END
```

〔2〕 演習問題2

(1) プログラムのフローチャート



(2) プログラムの解説

文番号	解説
1030～1040	諸条件の設定，流量 Q，重力加速度 G，河幅 B，河床勾配 SI，上流端水深 $H1$，上流端河床高 $Z1$，打切り誤差 E，断面間距離 DX
1050	① 式中の定数 AA，BB を求めている．($AA\to C1$，$BB\to C2$)
1060	上下流断面の河床高の差を求めている．
1090	断面数の設定．
1110	下流断面の河床高の計算および下流水深 h_2 の初期値設定．(上流断面の水深を与えている)
1120	① 式中の定数 CC を求めている．($CC\to C3$)
1130	① 式により，$f(h_2)$ を求めている．{$f(h_2)\to F$}
1140	収束の判定． $\|f(h_2)\|<\varepsilon$ の場合→収束したと判断し，文番号 1170 へ進む． $\|f(h_2)\|>\varepsilon$ の場合→収束していないので次の行へ進む．
1150～1152	$f'(h_2)$ および $f''(h_2)$ を求めている．{$f'(h_2)\to FD$，$f''(h_2)\to FDD$} 1階のニュートン法の場合は $f'(h_2)$ のみを計算．
1154～1060	Δh の計算（1階のニュートン法の場合は ⑮ 式，2階のニュートン法の場合は ⑯ 式）および h_2 の更新を行う．また繰返し回数を1回増して，文番号 1130 へもどる．この作業は収束するまで行う．
1170	計算断面の移動を行う．(下流河床高と計算された下流水深を，それぞれ上流の値におき換える)
1180	計算結果出力．
1190	文番号 1090 へもどり，次の断面の計算を行う。ただし，文番号 1090 で設定した断面数すべてが計算された場合は，文番号 1200 へ進み，計算終了となる．

(3) プログラムのリスト

```
1000 ' 射流の不等流計算プログラム  （広矩形断面）
1010 '   *****  1階の Ｎｅｗｔｏｎ法   *****
1020 CLEAR : DEFINT I-N
1030 Q=2000 : SN=.02 : G=9.8 : B=200 : SI=1/100 : H1=1.4
1040 Z1=50  : E=1E-03 : DX=100
1050 C1=Q^2/(2*G*B^2) : C2=DX*(SN*Q)^2/(2*B^2)
1060 DZ=DX*SI
1070  LPRINT "      NO   距離   河床高   水深    水位   繰返し回数"
1080  LPRINT "             (m)    (m)    (m)    (m)"
1090 FOR I=1 TO 51
1100  IF I=1 THEN 1180
1110  Z2=Z1-DZ : H2=H1 : K=1
1120  C3=Z2-(Z1+H1+C1/H1^2-C2/H1^(10!/3!))
1130   F=H2+C1/H2^2+C2/H2^(10!/3!)+C3
1140   IF ABS(F) < E THEN 1170
1150   FD=1-2*C1/H2^3-10/3*C2/H2^(13!/3!)
1160   DH=-F/FD : H2=H2+DH : K=K+1 : GOTO 1130
1170  H1=H2 : Z1=Z2
1180  LPRINT USING "      ##   ####    ##.#   #.###   ##.###      ##"
     ;I;(I-1)*DX;Z1;H1;Z1+H1;K
1190 NEXT I
1200 END
```

```
1000 ' 射流の不等流計算プログラム  （広矩形断面）
1010 '   *****  2階の Ｎｅｗｔｏｎ法   *****
1020 CLEAR : DEFINT I-N
1030 Q=2000 : SN=.02 : G=9.8 : B=200 : SI=1/100 : H1=1.4
1040 Z1=50  : E=1E-03 : DX=100
1050 C1=Q^2/(2*G*B^2) : C2=DX*(SN*Q)^2/(2*B^2)
1060 DZ=DX*SI
1070  LPRINT "      NO   距離   河床高   水深    水位   繰返し回数"
1080  LPRINT "             (m)    (m)    (m)    (m)"
1090 FOR I=1 TO 51
1100  IF I=1 THEN 1180
1110  Z2=Z1-DZ : H2=H1 : K=1
1120  C3=Z2-(Z1+H1+C1/H1^2-C2/H1^(10!/3!))
1130   F=H2+C1/H2^2+C2/H2^(10!/3!)+C3
1140   IF ABS(F) < E THEN 1170
1150   FD=1-2*C1/H2^3-10/3*C2/H2^(13!/3!)
1152   FDD=6*C1/H2^4+130/9*C2/H2^(16!/3!)
1154   DH=(-FD-SQR(FD*FD-2*FDD*F))/FDD
1160   H2=H2+DH : K=K+1 : GOTO 1130
1170  H1=H2 : Z1=Z2
1180  LPRINT USING "      ##   ####    ##.#   #.###   ##.###      ##"
     ;I;(I-1)*DX;Z1;H1;Z1+H1;K
1190 NEXT I
1200 END
```

〔3〕 演習問題3

(1)~1 プログラムのフローチャート

・1階のニュートン法



(2)~1 プログラムの解説

・1階のニュートン法

文番号	解説
30	断面数 NN
40	配列の宣言
50	追加距離 $D(I)$ の読込み(単位 m), ただし, Iは断面番号を表わし, 下流から順に 1~NN の値をとる
60	河床高 $Z(I)$ の読込み (単位 m)
70	河幅 $B(I)$ の 読込み (単位 m)
80	区間距離 $DL(I)$ の計算 (単位 m)
100	追加距離 $D(I)$ の DATA 文
110	河床高 $Z(I)$ の DATA 文
120	河幅 $B(I)$ の DATA 文
140	流量 Q の設定 (単位 m^3/sec)
150	粗度係数 N の設定 (単位 $m^{-\frac{1}{3}}\cdot sec$)
160	第1断面水深 $SH(1)$ の設定 (単位 m)
170	第1断面水位 $HH(1)$ の設定 (単位 m)
180	打切り誤差 EP の設定
190	重力加速度 G (単位 m/sec^2)
230	$J=1$ のとき, 文番号 340 へ
240	水深の初期値 HH の設定
250	係数 A の計算 ②式の AA
260	係数 B の計算 ③式の BB
270	係数 C の計算 ④式の CC
280	$f(HH)$ の計算
290	$f'(HH)$ の計算
300	$\Delta h=-f(HH)/f'(HH)$ の計算
310	$\|f(HH)\|<EP$ であれば文番号 330 へ
320	HH に Δh を加え文番号 280 へ
330	水深 $SH(J)$, 水位 $HH(J)$ の設定
340	結果の出力
350	$J=J+1$ とし, 次の断面へ
360	計算時間の出力
370	収束回数の出力

(3)~1 プログラムのリスト

・1階のニュートン法

```
20 TIME$="00/00/00"
30 NN=10
40 DIM D(NN),DL(NN),Z(NN),B(NN),SH(NN),HH(NN)
50 FOR I=1 TO NN:READ D(I):NEXT I          :'追加距離の読み込み
60 FOR I=1 TO NN:READ  Z(I):NEXT I         :'河床高の読み込み
70 FOR I=1 TO NN:READ  B(I):NEXT I         :'河幅の読み込み
80 FOR I=2 TO NN:DL(I)=D(I)-D(I-1):NEXT I  :'区間距離の計算
90 '
100 DATA 0,500,1000,1200,1800,2100,2500,3000,3300,3800 :'追加距離 (M)
110 DATA 0,.5,.9,.8,2.0,2.3,3.0,3.0,3.5,4.0           :'河床高   (M)
120 DATA 300,320,280,250,300,300,320,350,300,250       :'河幅     (M)
130 '
140 Q=1500            :'流量  M^3/S
150 N=.025            :'粗度
160 SH(1)=2.5         :'第一断面水深
170 HH(1)=SH(1)+Z(1):'第一断面水位
180 EP=1E-03          :'打ち切り誤差
190 G=9.8             :'重力加速度  M/SEC^2
200 '
205 LPRINT "ーーー広矩形断面における不等流計算ーーー"
206 LPRINT "NEWTON-RAPHSON 法を用いた水位の同定  微分項--->第1項目まで"
207 LPRINT
210 LPRINT "NO  追加距離  区間距離  河  幅  河床高  水深   水位  収束回数"
220 FOR J=1 TO NN:SN=0
230   IF J=1 THEN 340
240   HH=SH(J-1)
245  '********************** 係数A,B,Cの計算  *********************
250   A=Q*Q/(2*G*B(J)*B(J))
260   B=-Q*Q*N*N*DL(J)/(2*B(J)*B(J))
270   C=Z(J)-(Z(J-1)+SH(J-1)+Q*Q/(2*G*B(J-1)*B(J-1)*SH(J-1)*SH(J-1))+
        Q*Q*N*N*DL(J)/(2*B(J-1)*B(J-1)*SH(J-1)^(10/3)))
275 ' ********************* F(HH),F'(HH)の計算 ********************
280   FH=HH+A/(HH^2)+B/(HH^(10/3))+C        :'  F(HH)
290   F1=1-2*A/(HH^3)-10*B/(3*HH^(13/3))    :'  F'(HH)
300   DH=-FH/F1                             :'  ΔH
310   IF ABS(FH)<EP THEN 330                :'  条件判断
320   HH=HH+DH:SN=SN+1:GOTO 280             :'  HH=HH+DH
330   SH(J)=HH:HH(J)=SH(J)+Z(J)
340 LPRINT USING "##   ####.##   ####.##   ###.#  ##.####  ##.####
      ##.####  ##";J,D(J),DL(J),B(J),Z(J),SH(J),HH(J),SN:TSN=TSN+SN
350 NEXT J :LPRINT
360 LPRINT "  計算時間  (SEC) =";TIME$
370 LPRINT "  収束回数計       =";TSN
```

(1)～2　プログラムのフローチャート

・2階のニュートン法



(2)～2　プログラムの解説

・2階のニュートン法

文番号	解　　説
30	断面数 NN
40	配列の宣言
50	追加距離　$D(I)$ の読込み (単位 m)，ただし，I は断面番号を表わし，下流から順に 1～NN の値をとる
60	河 床 高　$Z(I)$ の読込み (単位 m)
70	河　　幅　$B(I)$ の読込み (単位 m)
80	区間距離　$DL(I)$ の計算 (単位 m)
100	追加距離　$D(I)$ の DATA 文
110	河 床 高　$Z(I)$ の DATA 文
120	河　　幅　$B(I)$ の DATA 文
140	流　　量　Q の設定 (単位 m^3/sec)
150	粗度係数　N の設定 (単位 $m^{-\frac{1}{3}}\cdot sec$)
160	第1断面水深　$SH(1)$ の設定 (単位 m)
170	第1断面水位　$HH(1)$ の設定 (単位 m)
180	打切り誤差　EP の設定
190	重力加速度　G　(単位 m/sec^2)
260	$J=1$ のとき，文番号 440 へ
270	水深の初期値　HH の設定
290	係数　A の計算　②式の AA
300	係数　B の計算　③式の BB
310	係数　C の計算　④式の CC
330	$f(HH)$ の計算
340	$f'(HH)$ の計算
350	$f''(HH)$ の計算
360	判別式 HT の計算
370	HT が負のとき，文番号 380 へ
	HT が正のとき，文番号 390 へ
380	$\Delta h=-f(HH)/f'(HH)$ の計算
	文番号 400 へ
390	$\Delta h=(-f'(HH)+\sqrt{HT})/f''(HH)$ の計算
400	$\|f(HH)\|<EP$ であれば，文番号 430 へ
410～420	HH に Δh を加え，文番号 330 へ
430	水深 $SH(J)$，水位 $HH(J)$ の設定
440	結果の出力
450	$J=J+1$ とし，次の断面へ
460	計算時間　TIME$ の出力
470	収束回数計　TSN の出力

(3)~2 プログラムのリスト

・2階のニュートン法

```
20 TIME$="00/00/00"
30 NN=10
40 DIM D(NN),DL(NN),Z(NN),B(NN),SH(NN),HH(NN)
50 FOR I=1 TO NN:READ D(I):NEXT I          :'追加距離の読み込み
60 FOR I=1 TO NN:READ  Z(I):NEXT I         :'河床高の読み込み
70 FOR I=1 TO NN:READ  B(I):NEXT I         :'河幅の読み込み
80 FOR I=2 TO NN:DL(I)=D(I)-D(I-1):NEXT I  :'区間距離の計算
90 '
100 DATA 0,500,1000,1200,1800,2100,2500,3000,3300,3800 :'追加距離  (M)
110 DATA 0,.5,.9,.8,2.0,2.3,3.0,3.0,3.5,4.0            :'河床高    (M)
120 DATA 300,320,280,250,300,300,320,350,300,250       :'河幅      (M)
130 '
140 Q=1500            :'流量  M^3/S
150 N=.025            :'粗度
160 SH(1)=2.5         :'第一断面水深
170 HH(1)=SH(1)+Z(1):'第一断面水位
180 EP=1E-03          :'打ち切り誤差
190 G=9.8             :'重力加速度  M/SEC^2
200 '
210 LPRINT "ーーー広矩形断面における不等流計算ーーー"
220 LPRINT "NEWTON-RAPHSON 法を用いた水位の同定  微分項--->第2項目まで"
230 LPRINT
240 LPRINT "NO  追加距離  区間距離  河  幅  河床高  水深    水位  収束回数"
250 FOR J=1 TO NN:SN=0
260   IF J=1 THEN 440
270   HH=SH(J-1)
280  '********************* 係数A,B,Cの計算  *********************
290   A=Q*Q/(2*G*B(J)*B(J))
300   B=-Q*Q*N*N*DL(J)/(2*B(J)*B(J))
310   C=Z(J)-(Z(J-1)+SH(J-1)+Q*Q/(2*G*B(J-1)*B(J-1)*SH(J-1)*SH(J-1))+
        Q*Q*N*N*DL(J)/(2*B(J-1)*B(J-1)*SH(J-1)^(10/3)))
320   **************** F(HH),F'(HH),F''(HH)の計算 ****************
330   FH=HH+A/(HH^2)+B/(HH^(10/3))+C         :'  F(HH)
340   F1=1-2*A/(HH^3)-10*B/(3*HH^(13/3))     :'  F'(HH)
350   F2=6*A/(HH^4)+130/9*B/HH^(16/3)        :'  F''(HH)
360   HT=F1^2-2*F2*FH
370   IF HT<0 THEN 380 ELSE 390              :'  HT<0 の場合,一次微分項
380   DH=-FH/F1:GOTO 400                     :'  までを採用
390   DH=(-F1+SQR(HT))/F2                    :'  ΔH
400   IF ABS(FH)<EP THEN 430                 :'  条件判断
410   HH=HH+DH:SN=SN+1                       :'  HH=HH+ΔH
420   GOTO 330
430   SH(J)=HH:HH(J)=SH(J)+Z(J)
440 LPRINT USING "##   ####.##   ####.##   ###.#  ##.####  ##.####
        ##.####  ##";J,D(J),DL(J),B(J),Z(J),SH(J),HH(J),SN:TSN=TSN+SN
450 NEXT J :LPRINT
460 LPRINT "  計算時間  (SEC)=";TIME$
470 LPRINT "  収束回数計      =";TSN
```